
               Processing Your Thoughts with 1-2-3
     (PC Magazine Vol 4 No 10 May 14, 1985 by Robert Flast)

     If you are intrigued by "thought processor" programs that help
you organize your ideas into a structured outline, you can use 123's
macros with its standard spreadsheet and data base functions to create
outlines similar to those generated by Framework, MaxThink, and
ThinkTank.
     The 123 "thought processor" allows you to numerically code your
ideas into four levels by using decimal numbers similar to AA.BBCCDD.
AA, BB, CC, and DD are numbers that range from 00 to 99.  AA represents
the major categories in your outline; BB indicates the subcatergories
within each major category; and CC and DD are lower subcategorizations.
You assign the category number to each idea, and the macro puts in the
outline numbers for you.  It starts your outline with 50.000000 instead
of 00.000000, so that you can later insert material at the beginning if
you find it necessary.  It calculates line numbers within the outline
by averaging the line above and the line below.  Changing the outline
means changing the coding.  While this coding technique is easy to
understand, you probably wouldn't find it simple to write codes and
continually change them while developing your outline.  Fortunately,
the macros do all the bookkeeping involved in adding or inserting lines.
The initizalization macro is in cells B4..B7.  The add macro, which
controls the insertion of lines, is in B10..B40.  The macro also
contains some formatting material designed to help you keep track of
the levels in your outline.
     The add macro numbers each row, or idea, according to the level
number that you input to indicate the location of the line.  To insert
a line, you first place the cursor on the line that's two lines below
the line you wish to insert and in the column that represents the level
of the thought in the outline.  You type Alt-A to invoke the add macro,
answer the request for the level number (which governs both the
numbering and indentation), and then type your line.
     If you choose, for instance, level 4, the macro will verigy that
the level-4 number above the insertion point is in fact different from
the level-4 number below the point.  If the levels above and below are
the same, the macro does nothing; if they are different, however, it
calculates a level-4 number for the new line that is between the upper
and lower numbers.  The macro then moves the cursor four columns to the
right of column A and waits for you to type out your line.
     After you have type in the macros, you position line A42 at the
top left-hand corner of the screen, move the cursor to A45, and type
/WTH to set the horizontal titles in rows 42 through 44 at the top of
the screen.  This command protects the macro range above your work
area.  You label the macro ranges by typing /RNC, \A B40 for the add
macro range and /RNC, \I and B4..B7 for the initialization macro range.
Now, you should save the worksheet, and you are ready to being using
this outline generator.
     To organize an article using these macros, you must first produce
a starting outline, including an introduction, a main body, and a
conclusion sectino, and number the main topics 50, 70, and 80.  You
type Alt-I to invoke the initialization macro and enter your first
thought.  To add your next though, you type Alt-A to use the add macro
for each line that you want to add to the document.  For example, to
add the conclusino heading, you place your cursor in the first level
column, where you type "main body" but two rows below (the macro
double-spaces your outline), and type Alt-A.  Because the conclusion
heading has to be in the same column as the main-body heading, you
enter 1 when prompted for the level number of the new line.  The macro



then shifts the existing text down two rows, calculates the new row
number, and positions the cursor at the appropriate level or column,
where you type the word "Conclusion."
     Next, you produce the outline within each main topic and assign a
number that corresponds to BB in the coding formula to each new thought
by positioning the cursor one column (which is equal to one level) to
the right and two rows below.  You then type Alt-A, the level number,
and the contents for each new entry.  If necessary, you continue this
process to the third and fourth levels; at that point, the outline
should be practically finished and you would just need to expand or
edit your thoughts.
     Even though the macro automatically moves the cursor to any column
you want via the level number that you enter, it cannot check its work
to make sure it is correct.  Therefore, you should use both the level-
numbering and the reinforcing indentation approaches to keep track of
the underlying structure of your thoughts.
     To move around the outline, you use the up, down, and page-move
cursor keys.  However, to move to a particular section, you have to use
the database query command (/DQ) to locate a particular level code.  If
you wish to reorganize your thoughts, you change the numerical codes to
reflect the desired sequence and sort the rows by using the /DS command
with column A as the primary key.
     You print your final results with the level codes and the titles
by printing teh range A45..Hxxx (xxx stands for the last line in the
outline).  If you do not want to see the underlying structure, you just
print the range B45..Hxxx.
     Outlining doesn't work for every person or every application, but
if you are an outliner by nature or are intrigued by the technique, you
should try to use these macros.  If you like the outlining process, you
may want to graduate to a dedicated outline processor.  But if the
concept doesn't suit your style, you haven't wasted your money on a
useless software program.

      A           B           C           D           E
-----------------------------------------------------------------
            THIS IS FOR INITIALIZING THE WORKSHEET
 4  \I  {HOME}0~/RFF6~~{RIGHT}'USE THIS AS A DUMMY STARTING LINE{LEFT}
 5      /RNCB~~{DOWN}{DOWN}{DOWN}{DOWN}/RNCT~~90~/RFF6~~
 6      {RIGHT}USE THIS AS A DUMMY EDING LINE{LEFT}{UP}{UP}50~/RFF6~~
 7      {RIGHT}/RNCH~{BS}~/XLENTER FIRST LINE HERE --->~H~
 8      /XQ
 9          THIS IS FOR INSERTING THE NECESSARY LINES
10  \A  {DOWN}{DOWN}/WIR{DOWN}{LEFT}~{LEFT}{UP}{UP}
11      /RNDB~/RNCB~~{DOWN}{DOWN}{DOWN}{DOWN}/RNDT~/RNCT~~{UP}{UP}
12      /XNENTER LEVEL NUMBER (1/2/3/4)~F12~
13      /XGB15~
14      /XNWRONG LEVEL.  RE-ENTER NUMBER (1/2/3/4)~F12~
15      /XIF12<>1~/XGB20~
16      /XI@INT(B)=@INT(T)~/XGB14~
17      @INT((B+T)/2)~/RFF6~~
18      {RIGHT}/RNDH~/RNCH~~/XLENTER LINE HERE --->~H~
19      /XQ
20      /XIF12<>2~/XGB27~
21      /XI@INT(B*100)=@INT(T*100)~/XGB14~
22      XI@INT(B)<>@INT(T)~/XGB24~
23      @INT((@INT(B*100)+@INT(T*100))/2)/100~/RFF6~~/XGB25~
24      @INT((@INT(B*100)+(@INT(B)*100+99))/2)/100~/RFF6~~
25      {RIGHT}{RIGHT}/RNDH~/RNCH~~/XLENTER LINE HERE --->~H~{LEFT}
26      /XQ



27      /XIF12<>3~/XGB34~
28      /XI@INT(B*10000)=@INT(T*10000)~/XGB14~
29      /XI@INT(B*100)<>@INT(T*100)~/XGB31~
30      @INT((@INT(B*10000)+@INT(T*10000))/2)/10000~/RFF6~~/XGB32~
31      @INT((@INT(B*10000)+(@INT(B*100)*100+99))/2)/10000~/RFF6~~
32      {RIGHT}{RIGHT}{RIGHT}/RNDH~/RNCH~~/XLENTER LINE HERE --->~H~{LEFT}{LEFT}
33      /XQ
34      /XIF12<>4~/XGB14~
35      /XI@INT(B*1000000)=@INT(T*1000000)~/XGB40~
36      /XI@INT(B*10000)<>@INT(T*10000)~/XGB38~
37      @INT((@INT(B*1000000)+@INT(T*1000000))/2)/1000000~/RFF6~~/XGB39~
38      @INT((@INT(B*1000000)+(@INT(B*10000)*100+99))/2)/1000000~/RFF6~~
39      {RIGHT}{RIGHT}{RIGHT}{RIGHT}/RNDH~/RNCH~~/XLENTER LINE HERE --->~H~{LEFT}
{LEFT}{LEFT}
40      /XQ
41
42  LEVEL=> FIRST       SECOND      THIRD       FOURTH
43  CODE    AA.000000   AA.BB0000   AA.BBCC00   AA.BBCCDD
44  ----------------------------------------------------------------------
45
###################################################################################
###################################################################################
##################################################################


